Population Dynamics

a) Prove that in the Volterra-Lotka predator-prey model the number of the animals changes periodically which means that the trajectories drawn in the phase lane are closed curves.

b) Prove that such a transformation of the variables can be found for the Volterra-Lotke system of differential equation with which the initial differential equation can be transformed to a Hamilton differential equation.

Consider a biological micro environment in which two species live. Let's call one of the species predator and the other prey. Naturally, the predators eat the preys, which raises the challenging question that how the number of the population changes as time passes. Let's see the further properties of the model.

1. If predators didn't exist then the number of the preys would exponentially rise. Let's suppose that there is enough food for the preys in their environment.

2. If preys didn't exist then the number of the predators would exponentially decrease due to the lack of food.

3. The phenomenon of cycle. If the number of the predators increases for a pro tem basis then the number of the preys decreases because more predators eat more. However, if the number of the preys decreases then the amount of food decreases as well so after a while the number of the predators will do so. But less predators eat less so the number of the preys starts rising because there is nobody to eat them. Finally, if the number of the preys increases then the number of the predators also rises because they will have much more food. And then the cycle starts again.

In this model the predators and the preys live next to each other and although their numbers change dynamically none of the species disappears. We are going to show that the following system consisting of two nonlinear differential equations, which is called Volterra-Lotka predator-prey model, satisfies the strict conditions mentioned above.

> *restart*

> $$Volterra\_Lotka := \left[\frac{d}{dt}\, x(t) = x(t)\,(a - b\,y(t)),\ \frac{d}{dt}\, y(t) = y(t)\,(-c + d\,x(t)) \right]$$

$$Volterra\_Lotka := \left[\frac{d}{dt}\, x(t) = x(t)\,(a - b\,y(t)),\ \frac{d}{dt}\, y(t) = y(t)\,(-c + d\,x(t)) \right] \qquad \textbf{(1)}$$

The meanings of the variables used in the model are the following:

• x(t)

the number of the preys at the moment of t is $(0<x(t))$

• y(t): the number of the predators at the moment of t is $(0<y(t))$

• a: the parameter that determines the rise of the prey. If there weren't any predators then their rise would be

• c: the parameter that determines the decrease of the predators. If there weren't any preys then their decrease would be [képlet]

• b: the parameter that determines the decrease of the prey, which depends on how often the prey and the predators meet.

• d: the parameter that determines the rise of the predators, which depends on how often the prey and the predators meet.

Let's write the a, b, c and d parameters in the model and plot the direction field.

> $a, b, c, d := 3, 2, 5, 4$

$$a, b, c, d := 3, 2, 5, 4 \quad (2)$$

> $with(DEtools)$:

```
> dfieldplot(Volterra_Lotka, [x(t),y(t)], t =0..2, x=0..3, y=0..3,
  arrows=THICK, scaling=constrained,title="Vektormezö a Volterra-
  Lotka modellhez");
```



Now let's draw some solutions in the (x,y) phase plane.

```
> DEplot(Volterra_Lotka, [x(t),y(t)], t=0..3, {[0,.2,1.5],[0,.4,
  1.5], [0,.6,1.5], [0,.8,1.5], [0,1,1.5], [0,1.2,1.5], [0,1.25,
  1.5]}, stepsize=0.01, scene=[x,y], scaling=constrained,
  linecolor=blue, thickness=2, title="A Volterra-Lotka modell
```

```
fázissíkja");
```



The closed curves received in the phase plane suggest that the solutions are periodical functions which means that the $[x(T), y(T)] = [x(0), y(0)]$

equality is fulfilled for every [x,y] solution for some kind of 0<T value. So after a while the number of the animals in the population coincides with the initial value. Naturally different T periodical time belongs to different solutions.

We can see from the position of the arrows of the vector field that the closed

$[x(t), y(t)]$ trajectories get around the

$$X = \frac{4}{5}, Y = \frac{3}{2}$$

balance state counter clockwise.

The periodicity mentioned above means that neither of the species disappears in the model. It is known that the trajectories are isolated in an autonomous system, that is, the curves can cross neither each other nor themselves. Furthermore, it is also obvious that the x(t) 0 and the y(t)=0 axes are solutions to the system of the differential equation. So we can conclude that the arbitrary solution originating from the first quarter always stays at the first quarter. Naturally, if a solution gets very far away from the [X,Y] balance state then the sway of the system will be big and it can happen that the number of one of the species decreases to 0.

After such a long preparation let's try to prove the suggested periodicity property of the solutions with analytic tools.

The solutions of the system of differential equation cannot be created explicitly but the first integral of the system of the differential equation can be determined by clever transformations. Namely we can find an implicit equation between the x and y by substituting the solutions into which we can get a constant value independent of t.

To prove this, free the a, b, c and d parameters and keep in mind that the x and y are the functions of the t parameter.

> $a := 'a' :\quad b := 'b' :\quad c := 'c' :\quad d := 'd' :$

Divide the first equation by x(t) and the second by y(t).

> $\dfrac{Volterra\_Lotka_1}{x(t)}$

$$\frac{\frac{\mathrm{d}}{\mathrm{d}t}\, x(t)}{x(t)} = a - b\, y(t) \tag{3}$$

> $\dfrac{Volterra\_Lotka_2}{y(t)}$

$$\frac{\frac{\mathrm{d}}{\mathrm{d}t}\, y(t)}{y(t)} = -c + d\, x(t) \tag{4}$$

> **(3)** $(-c + d\, x(t))$

$$\frac{(-c + d\, x(t))\left(\frac{\mathrm{d}}{\mathrm{d}t}\, x(t)\right)}{x(t)} = (-c + d\, x(t))\,(a - b\, y(t)) \tag{5}$$

> **(4)** $(a - b \cdot y(t))$

$$\frac{(a - b\, y(t))\left(\frac{\mathrm{d}}{\mathrm{d}t}\, y(t)\right)}{y(t)} = (-c + d\, x(t))\,(a - b\, y(t)) \tag{6}$$

> $de :=$ **(5)** $-$ **(6)**

$$de := \frac{(-c + d\, x(t))\left(\frac{\mathrm{d}}{\mathrm{d}t}\, x(t)\right)}{x(t)} - \frac{(a - b\, y(t))\left(\frac{\mathrm{d}}{\mathrm{d}t}\, y(t)\right)}{y(t)} = 0 \tag{7}$$

We multiplied the first equation by –c+dx and the second by a-by so that the right sides of the two equations should have the same syntaxes. After the extraction of the two equations we get the system of differential equation with a variable that can be separated in a parametric way. Integrate the differential equation by the t.

> $\int lhs(de)\, \mathrm{d}t = C$

$$d\, x(t) - c \ln(x(t)) + b\, y(t) - a \ln(y(t)) = C \tag{8}$$

The response is that the $[x(t), y(t)]$ satisfies the (8) implicit equation. And how can we create the following V function with two variables?

> $V := (x, y) \rightarrow d\,x - c\ln(x) + b\,y - a\ln(y)$

$$V := (x, y) \rightarrow d\,x - c\ln(x) + b\,y - a\ln(y) \quad \textbf{(9)}$$

The V function is the first integral of the differential equation. Although we do not know the solutions to the differential equation we know that the solutions are on the V(x,y)=C contours of the V function. In this way the examination of the V function can contribute to the determination of the periodicity properties of the solutions.

Let's look at the V function and its contours in the case of fixed a,b,c and d parameter values.

> $ei := subs(a = 3, b = 2, c = 5, d = 4, V(x, y))$ # *első integrál*

$$ei := 4\,x - 5\ln(x) + 2\,y - 3\ln(y) \quad \textbf{(10)}$$

```
> plot3d(ei, x =.1..3, y=.1..3, orientation=[58,105], style=patch,
  axes=boxed, title=`A V(x,y) mint elsö integrál`);
```

A V(x,y) mint els□ integr□l




```
> plots[contourplot](subs(a=3, b=2, c=5, d=4, V(s,t)), s=.1..3, t=
  .1..3, axes=normal, color=black, scaling =constrained, contours=
  [5.75,6,6.5,7,7.5,8,8.5, 9], title=`Az elsö integrál
```

```
szintvonalai`);
```



This graph reminds us of the plotting of the phase portrait. Let's prove that the $\left[X=\frac{c}{d}, Y=\frac{a}{b}\right]$ balance state is the strict minimum point of the first integral of the V. Break up the V first integral to the sum of two functions. The F should be the sum of the terms depending on the x and the G on the y.

> $x := 'x'; y := 'y'$

$$x := x$$
$$y := y \tag{11}$$

> $F := unapply(op(1, V(x, y)) + op(2, V(x, y)), x)$

$$F := x \rightarrow d\,x - c\ln(x) \tag{12}$$

> $G := unapply(op(3, V(x, y)) + op(4, V(x, y)), y)$

$$G := y \rightarrow b\,y - a\ln(y) \tag{13}$$

> $'V(x, y) = F(x) + G(y)' = is(V(x, y) = F(x) + G(y))$

$$(V(x, y) = F(x) + G(y)) = true \tag{14}$$

The third command proves that the $V(x, y) = F(x) + G(y)$ equality is true. Let's see the F function first. Let's start with the graph of the function.

```
> plot(subs(c=5, d=4, F(x)), x=0..4.5, y=0..10, title=`F függvény`)
  ;
```



F has a strict extremum at the $X = \frac{c}{d}$ as we can see from the Maple instructions below.

> $\frac{d}{dx}\ 'F(x)' = D(F)$

$$\frac{d}{dx} F(x) = \left(x \to d - \frac{c}{x}\right) \quad \textbf{(15)}$$

> $minimum\_x := solve((rhs(\%))(x) = 0, x)$

$$minimum\_x := \frac{c}{d} \quad \textbf{(16)}$$

> $\frac{d^2}{dx^2}\ 'F(x)' = (D^{(2)})(F)$

$$\frac{d^2}{dx^2} F(x) = \left(x \to \frac{c}{x^2}\right) \quad \textbf{(17)}$$

> $0 < (rhs(\%))(minimum\_x)$

$$0 < \frac{d^2}{c} \quad (18)$$

Let's examine the G function the same way. These are going to be showed without any further comments.

```
> plot(subs(a=3, b=2, G(y)), y= 0.01..6, title=`G fuggveny`);
```



> $\frac{d}{dy}\, 'G(y)' = D(G)$

$$\frac{d}{dy} G(y) = \left(y \rightarrow b - \frac{a}{y}\right) \quad (19)$$

> $minimum\_y := solve((rhs(\%))(y), y)$

$$minimum\_y := \frac{a}{b} \quad (20)$$

> $\frac{d^2}{dy^2}\, 'G(y)' = (D^{(2)})(G)$

(21)

$$\frac{d^2}{dy^2} G(y) = \left(y \rightarrow \frac{a}{y^2} \right) \quad \textbf{(21)}$$

> $0 < (rhs(\%))(minimum\_y)$

$$0 < \frac{b^2}{a} \quad \textbf{(22)}$$

Because the $V(x, y) = F(x) + G(y)$, $X = \frac{c}{d}$ and the F have a strict absolute minimum at the $Y = \frac{a}{b}$ and the G has a strict absolute minimum at the$[X, Y]$

Based upon the proved strict minimum properties of the V we can state that its contours, on which the solutions of the Volterra-Lotka equation lie, are closed curves. For the periodicity of the solutions of the Volterra-Lotka equation we only have to prove that the solutions completely fill the lines of the V, which is ensured by the Poincare-Bendixon theorem that describes the solutions of the autonomous system of differential equation in the plane.

We have solved the a) part of the task. To solve the b) part of the task let's try the following substitution.

> $transzformacio := \{x(t) = e^{p(t)}, y(t) = e^{q(t)}\}$

$$transzformacio := \{x(t) = e^{p(t)}, y(t) = e^{q(t)}\} \quad \textbf{(23)}$$

Let's calculate how the Volterra-Lotka differential equations are transformed if we change from the previous variables [képlet] to the new ones [képlet]. The dchangevar procedure of the PDEtools package calculates this certain differential equation after the transformation.

> $tr\_de := PDEtools_{dchange}(transzformacio, Volterra\_Lotka, [p(t), q(t)])$

$$tr\_de := \left[\left(\frac{d}{dt} p(t) \right) e^{p(t)} = e^{p(t)} \left(a - b\, e^{q(t)} \right), \left(\frac{d}{dt} q(t) \right) e^{q(t)} = e^{q(t)} \left(-c + d\, e^{p(t)} \right) \right] \quad \textbf{(24)}$$

The parameters of the call can be seen on the above call of the dchangevar procedure. Its first parameter contains the transformation equalities that describe the conversions. The second contains the differential equation and the third parameter contains the names of the dependent variables after the transformation.

Let's divide the two differential equations by

$e^{p(t)}$and $e^{q(t)}$ Feel free to do this because both values are bigger than zero.

> $h_1 := \frac{tr\_de_1}{e^{p(t)}}$

$$h_1 := \frac{d}{dt} p(t) = a - b\, e^{q(t)} \quad \textbf{(25)}$$

> $h_2 := \frac{tr\_de_2}{e^{q(t)}}$

$$h_2 := \frac{d}{dt} q(t) = -c + d\, e^{p(t)} \quad \textbf{(26)}$$

The question is whether the right sides of the differential equations can be received as a partial derivative of a H=H(p,q), a so-called Hamilton function? The Hamilton function can be created from the V function as follows:

> $H := (s, t) \rightarrow V(e^s, e^t)$

$$H := (s, t) \rightarrow V(e^s, e^t) \quad (27)$$

> $H(p(t), q(t))$

$$d\, e^{p(t)} - c \ln(e^{p(t)}) + b\, e^{q(t)} - a \ln(e^{q(t)}) \quad (28)$$

> $Hamilton\_fuggveny := simplify(\%, symbolic)$

$$Hamilton\_fuggveny := d\, e^{p(t)} - c\, p(t) + b\, e^{q(t)} - a\, q(t) \quad (29)$$

We have got the Hamilton function by the simplification of the H(p,q).

After this, let's calculate the partial derivatives by the p and q of the H(p,q) expression. For this, the new [s,t] variables have to be introduced because the p and q are functions.

> $\frac{\partial}{\partial p} H(p,q) = subs(\{s = p(t), t = q(t)\}, \frac{\partial}{\partial s} H(s, t))$

$$\frac{\partial}{\partial p} H(p,q) = -c + d\, e^{p(t)} \quad (30)$$

> $\frac{\partial}{\partial q} H(p,q) = subs\left(\{s = p(t),\ t = q(t)\},\ \frac{\partial}{\partial t} H(s, t)\right)$

$$\frac{\partial}{\partial q} H(p,q) = b\, e^{q(t)} - a \quad (31)$$

We get the first Hamilton equation from the sum of the h1 and the (31) after a minor conversion. The same is true for the h2 and (30) expressions.

> $h_1 +$ **(31)**

$$\frac{d}{dt} p(t) + \frac{\partial}{\partial q} H(p,q) = 0 \quad (32)$$

> $Hamilton_1 := isolate\left(\%, \frac{\partial}{\partial t} p(t)\right)$

$$Hamilton_1 := \frac{d}{dt} p(t) = - \frac{\partial}{\partial q} H(p,q) \quad (33)$$

> $h_2 -$ **(30)**

$$\frac{d}{dt} q(t) - \frac{\partial}{\partial p} H(p,q) = 0 \quad (34)$$

> $Hamilton_2 := isolate\left(\%, \frac{\partial}{\partial t} q(t)\right)$

$$Hamilton_2 := \frac{d}{dt} q(t) = \frac{\partial}{\partial p} H(p,q) \quad (35)$$

>

We have solved the task.

What Have You Learnt About Maple?

- The contours of the functions with two variables can be plotted with the contourplot procedure of the plots package. Its syntax is:

$$contourplot(kifejezés\ x,\ y\text{–}ra,\ x = a\,..b,\ y = c\,..d,\ további\ opciók),$$

It draws the contours by the [a,b]x[c,d] rectangle and returns a 3D plot structure as a result.

- With the help of the dchangevar procedure of the PDEtools package the independent and dependent variables located in the system of the differential equation can be substituted with new variables. The dchangevar calculates the new system of differential equation received after the changing of the variables. Its syntax is

$$dchangevar(helyettesítő\ egyenletek,\ differenciálegyenletek,\ új\ függő\ változók).$$

Exercises

1.

Prove the property called Volterra principle of the Volterra-Lotka differential equation with the help of Maple.

If [képlet] is the arbitrary solution to the Volterra-Lotka equation and 0<T denotes its periodic time then the following equalities are satisfied:

$$\frac{\int_0^T y(t)\,\mathrm{d}t}{T} = \frac{a}{b} \quad és \quad \frac{\int_0^T x(t)\,\mathrm{d}t}{T} = \frac{c}{d}\ .\square$$